HP42S Triangle Solutions

REVISED, 2ND EDITION





HP42S Triangle Solutions

REVISED, 2ND EDITION

published by Software by D'Zign

Tollhouse, California, U.S.A, 1994

ISBN 0-944889-13-1

TECHNICAL ASSISTANCE

The program material, instructions and procedures contained in this book assume that the user has a working knowledge of both surveying *and* the general operation of the HP-42S calculator.

Technical assistance is limited to verification of the results shown in the various examples used in this book.

If you have any questions or suggestions regarding this book, or other **D'Zign** publications, please feel free to call us. The number is (209) 297-8025, and someone is available to answer technical questions between the hours of ***8:00 A.M. and Noon***, (Pacific Time Zone), ***Monday through Thursday***

Before calling for help, take a look through *"The Most Commonly Asked Questions"*, on the inside of the back cover.

WARNING

D'Zign
P.O. Box 430
Tollhouse, CA 93667-0430

introduction

Hewlett-Packard has produced a really powerful calculator at a very good price, the HP-42 Scientific Calculator, which lends itself nicely to solving surveying problems. It can not be programmed by insertion of a surveying module, like the HP-41 or HP-48, but typing in a program has been made very simple.

the operations index

To find a function for the first time, HP has provided an "***Operations Index***", on pages 310 through 335 of the instruction manual, which tells you exactly what keystrokes to use to type in the function you want.

Even better, this index gives you the page number that you may refer to if you want to know more about the function you are using. If, while typing in a program, you aren't sure how to input a particular function, simply refer to the *Operations Index.*

the softkey menus

All of the programs in this booklet take advantage of the ***softkey*** menu system built into this calculator. When you want to start a program you stroke the XEQ key and then the softkey corresponding to the program you want, shown in the menu displayed at the bottom half of the screen.

learning while programming

We have tried to write this series of booklets in such a way that you quickly become aquainted with the calculator and its functions while you are programming it.

The *use* of the triangle program is simplicity itself. You need three known parts of the triangle being solved, and there are only three keys used for input. They are in the *menu* as SIDE, ANGLE and AREA, and to select them you stroke the key under the menu listing.

The only other keys used are one menu listing that returns you quickly to the program for another solution, and the R/S key.

The use of a printer is not required, but does make the job of proof reading easier, Hewlett-Packard has the InfraRed Printer available for the 42S, and one of the great features of this calculator is that it already has an InfraRed transmitter built in.

subroutines

Because of the way the calculator works, we will start with the input of some subroutines.

Once a subroutine has been input, its name appears in the menu when you stroke the **XEQ** key, and all you have to do to add it *as a program step* in another program is to stroke **GTO** or **XEQ**, followed by keystroking the key corresponding to the subroutine when the menu appears. You may scroll through the menus with the ▲ or ▼ key.

```
00 { 47-Byte Prgm }
01▶LBL "CL"
02 ΣREG 00
03 CLΣ
04 ΣREG 11
05 CLΣ
06 0
07 STO 24
08 RTN
09▶LBL "FCL"
10 0.013
11▶LBL "FN"
12 CF IND ST X
13 ISG ST X
14 GTO "FN"
15 ENTER
16 SF 21
17 END
```

getting started

Begin by stroking the shift key, then the **XEQ** key. The display will show a menu which will be blank (if you haven't yet input any programs) except for .END. , on the left. The keys just below each of the menu portions will correspond to the menu *instruction* above it.

Stroke the key just below the **.END.** in the display menu, and then go into ***program mode*** by stroking the *shifted* **R/S** key. Scroll upward once with the ▲ key.

```
00▶{ 0-Byte Prgm }
01 .END.
```

Your display should be similar to the one shown to the left. Begin typing in the program steps shown in the illustration above, LBL "CL".

Proof read the program. If you scroll to 00 it should now say "47-Byte Prgm" . . . the byte count is one quick check on the program, but you still need to check every step. Look, in particular, for steps that are **alpha** (with " " marks) but *shouldn't* be. Or should be but aren't.

"DMS"

This program puts the output of angles into the form ° ' ", and gives all answers to the nearest tenth of a second.

NOTE: Before beginning the input, review pages 130 and 131 of your manual, which tells you about the 'append' symbol (x) used in program steps 10, 19, 29 and 38. Keystrokes for the degree symbol are shown in the MATH section (menu maps, page 295), the " and ' on page 296 (the ' looks different in the keystroke than it does when displayed or printed . . the one you want is the one just above the ¿ symbol).

Repeat the procedure of going to the permanent **.END.**, scroll up to 00, and input the program below.

```
01▶LBL "DMS"   11 -          21 FP          31 FIX 04
02 FS?C 19     12 100        22 100         32 SF 29
03 CLA         13 ×          23 ×           33 RCL 19
04 ENTER       14 ABS        24 FIX 02      34▶LBL 01
05 STO 19      15 STO 18     25 RND         35 10
06 IP          16 IP         26 FIX 01      36 X<>Y
07 CF 29       17 XEQ 01     27 XEQ 01      37 X<Y?
08 FIX 00      18 ARCL ST X  28 ARCL ST X   38 ⊢"0"
09 ARCL ST X   19 ⊢"'"       29 ⊢"""        39 END
10 ⊢"""        20 RCL 18     30 CLX
```

The byte count should be 78 bytes. Read through the steps to check for typos, and as a final check, input a number (try 25.25252), set flag 19, and execute the *DMS* function. Stroke the *shifted* ENTER key, to enter alpha, and your display should show 25°25'25.2" above the menu bar. If it doesn't, check it again.

the main program

The next two pages contain the main program. Take your time typing it in, to avoid errors.

Don't type an **END** on this one, the permanent end will work just as well, and this removes it from your XEQ/GTO menu.

```
01▶LBL "TRI<"
02 XEQ "CL"
03 XEQ "FCL"
04 0.002
05 STO 00
06 CLX
07 XEQ 20
08 RTN
09▶LBL 21
10 FS? 09
11 SF 03
12 FS? 04
13 SF 01
14 SF 04
15 ISG 00
16 RTN
17 GTO 10
18 RTN
19▶LBL 22
20 FS? 01
21 SF 08
22 FS? 04
23 SF 06
24 FS? 01
25 CF 06
26 FS? 09
27 SF 05
28 SF 09
29 ISG 00
30 RTN
31 GTO 10
32▶LBL 23
33 SF 07
34 ISG 00
35 RTN
36▶LBL 10
37 FS? 07
38 GTO 06
39 FS? 08
40 GTO 15
41 FS? 06
42 GTO 07
43 FS? 03
44 GTO 12
45 GTO 11
46 RTN
47▶LBL 06
48 FS? 05
49 GTO 18
50 FS? 01
51 GTO 17
52 GTO 16
53 RTN
54▶LBL 07
55 FS? 01
56 GTO 14
57 FS? 03
58 GTO 12
59 GTO 13
60 RTN
61▶LBL 11
62 STO 04
63 R↓
64 STO 02
65 R↓
66 STO 09
67 R↓
68 R↓
69 +
70 +
71 2
72 ÷
73 STO 07
74 X↑2
75 LASTX
76 RCLX 02
77 -
78 RCL 09
79 RCLX 04
80 ÷
81 SQRT
82 ACOS
83 2
84 x
85 STO 05
86 SIN
87 RCLX 09
88 STO 08
89 RCL 07
90 X↑2
91 LASTX
92 RCLX 09
93 -
94 RCL÷ 02
95 RCL÷ 04
96 SQRT
97 ACOS
98 2
99 x
100 STO 03
101 RCL 05
102 XEQ 00
103 STO 01
104 GTO 01
105 RTN
106▶LBL 12
107 " 2nd Solution"
108 FS? 20
109 AVIEW
110 FS?C 20
111 STOP
112 →HR
113 STO 01
114 R↓
115 STO 09
116 R↓
117 →HR
118 STO 05
119▶LBL 03
120 RCL 01
121 XEQ 00
122 STO 03
123 RCL 05
124 RCL 09
125 →REC
126 X<>Y
127 STO 08
128 RCL 03
129 1
130 →REC
131 R↓
132 ÷
133 STO 02
134 R↓
135 x
136 +
137 STO 04
138 GTO 01
139▶LBL 13
140 →HR
141 STO 03
142 R↓
143 →HR
144 STO 01
145 R↓
146 STO 09
147 RCL 03
148 RCL 01
149 XEQ 00
150 RCL 09
151 RCL 01
152 XEQ 04
153 GTO 03
154▶LBL 14
155 STO 02
156 R↓
157 →HR
158 STO 01
159 R↓
160 STO 09
161 RCL 01
162 RCL 02
163 →REC
164 RCL- 09
165 →POL
166 STO 04
167 RCL 09
168 RCL 02
169 RCL 04
170 GTO 11
171▶LBL 15
172 →HR
173 STO 03
174 R↓
175 STO 02
176 R↓
177 STO 09
```

```
178 RCL 03
179 SIN
180 RCL× 02
181 RCL÷ 09
182 ASIN
183 STO 05
184 RCL 03
185 XEQ 00
186 STO 01
187 RCL 05
188 RCL 09
189 RCL 01
190 XEQ 04
191 XEQ 03
192 180
193 RCL- 05
194 RCL+ 03
195 +/-
196 180
197 +
198 →HMS
199 RCL 02
200 RCL 03
201 →HMS
202 SF 20
203 GTO 12
204▶LBL 00
205 +
206 COS
207 +/-
208 ACOS
209 RTN
210▶LBL 01
211 FS? 10
212 CF 02
213 FS? 20
214 CF 02
215 FS? 02
216 CLA
217 "Side 1 = "
218 ARCL 09
219 ⊢"↳"
220 RCL 01
221 →HMS
222 ⊢"Angle 1 = "
223 XEQ "DMS"
224 AVIEW
225 ADV
226 "Side 2 = "
227 ARCL 02
228 ⊢"↳"
229 RCL 03
230 →HMS
231 ⊢"Angle 2 = "
232 XEQ "DMS"
233 AVIEW
234 ADV
235 "Side 3 = "
236 ARCL 04
237 ⊢"↳"
238 RCL 05
239 →HMS
240 ⊢"Angle 3 = "
241 XEQ "DMS"
242 AVIEW
243 ADV
244 RCL 08
245 RCLx 04
246 2
247 ÷
248 "AREA = "
249 ARCL ST X
250 AVIEW
251 ADV
252 FS? 10
253 XEQ 08
254 RTN
255▶LBL 08
256 " 2nd Solution"
257 FS?C 10
258 AVIEW
259 STOP
260 180
261 RCL- 01
262 RCL 09
263 X<>Y
264 →HMS
265 RCL 02
266 GTO 14
267▶LBL 09
268 R↓
269 R↓
270 RTN
271▶LBL 04
272 STO 01
273 R↓
274 STO 09
275 R↓
276 STO 05
277 RTN
278▶LBL 16
279 →HR
280 STO 01
281 SIN
282 XEQ 02
283 STO 02
284 XEQ 05
285 RTN
286▶LBL 17
287 SF 10
288 STO 02
289 XEQ 02
290 ASIN
291 STO 01
292 XEQ 05
293 RTN
294▶LBL 18
295 →HR
296 STO 03
297 X<>Y
298 →HR
299 STO 01
300 +
301 180
302 X<>Y
303 -
304 STO 05
305 SIN
306 2
307 X
308 X
309 RCL 01
310 SIN
311 RCL 03
312 SIN
313 X
314 ÷
315 SQRT
316 STO 02
317 RCL 01
318 →HMS
319 X<>Y
320 RCL 03
321 →HMS
322 XEQ 12
323 RTN
324▶LBL 02
325 X<>Y
326 STO 09
327 X
328 ÷
329 2
330 X
331 RTN
332▶LBL 05
333 RCL 09
334 RCL 01
335 →HMS
336 RCL 02
337 XEQ 14
338 RTN
339▶LBL 20
340 CLA
341 CLMENU
342 "SIDE"
343 KEY 1 GTO 21
344 "ANGL"
345 KEY 2 GTO 22
346 "AREA"
347 KEY 3 GTO 23
348 "MORE"
349 KEY 6 GTO 24
350 MENU
351 RTN
352▶LBL 24
353 GTO "TRI<"
354 .END.
```

The triangle shown to the right will be used for the examples. It should be noted that the output will vary slightly, depending on the number of places input, particularly in the input of the angles.



The notations for angles and sides is familiar to HP users, but is not the standard, or *textbook* notation which you have learned in trigonometry (side **a** opposite angle **A**, side **b** opposite angle **B**, and side **c** opposite angle **C**). The sides and angles are numbered, in order, going around the triangle.



The example triangle (top) shows this style of labeling, compared to the standard notation for sides and angles. ***Side 1*** may be assigned to any side that is convenient to use, depending upon the available information about the triangle. It should be located at a side where the known information then falls into position for solution by one of the routines.



In the example, the assigned designations go clockwise, If it will better fit the information available, the labeling may go anticlockwise instead, as shown to the left.

NOTE! There is no solution for a triangle where the three angles are the only known parts, since this condition can produce an infinite number of similar triangles.

side 1, side 2, side 3

THREE SIDES KNOWN is one of the most used solutions for triangles, particularly in recent years in surveying.

The lower cost and higher accuracy of electronic distance measurement equipment has resulted in more trilateration being used, instead of time-consuming repetitions of the angles.

keystroke example:



output:

```
Side 1 = 8.3640
Angle 1 = 36°12'32.0"

Side 2 = 9.6800
Angle 2 = 59°19'11.8"

Side 3 = 5.7450
Angle 3 = 84°28'16.2"

AREA = 23.9138
```



This routine uses the equations

$$A_3 = 2\cos^{-1}\sqrt{\frac{P(P - S_2)}{S_1 S_3}}$$

$$A_2 = 2\cos^{-1}\sqrt{\frac{P(P - S_1)}{S_2 S_3}}$$

and

$$A_1 = \cos^{-1}(-\cos(A_3 + A_2))$$

where $P = \frac{1}{2}(S_1 + S_2 + S_3)$

To solve another triangle, stroke MORE after the output.

side 1, angle 1, side 2



TWO SIDES AND THE INCLUDED ANGLE KNOWN is resolved by finding the third side, and then solving the triangle as shown on the previous page. The third side is found through the use of the equation $S_3 = \sqrt{S_1^2 + S_2^2 - 2\,S_1 S_2 \cos A_1}$

keystrokes:

5 . 7 4 5 SIDE

8 4 . 2 8 1 8 ANGL

8 . 3 6 4 SIDE

output:

```
Side 1 = 5.7450
Angle 1 = 84°28'18.0"

Side 2 = 8.3640
Angle 2 = 36°12'31.4"

Side 3 = 9.6800
Angle 3 = 59°19'10.6"

AREA = 23.9138
```

NOTE! Areas are calculated by this program using the equation shown below.



Area = ½(S_1 S_3 sin A_3)

side 1, angle 1, angle 2



keystrokes:

5 . 7 4 5 SIDE

8 4 . 2 8 1 8 ANGL

3 6 . 1 2 3 ANGL

output:

```
Side 1 = 5.7450
Angle 1 = 84°28'22.7"

Side 2 = 8.3640
Angle 2 = 36°12'30.0"

Side 3 = 9.6802
Angle 3 = 59°19'07.3"

AREA = 23.9139
```

ONE SIDE AND THE TWO FOLLOWING ANGLES KNOWN.
This solution first solves for the third angle with the equation

$$A_3 = \cos^{-1}\left(-\cos\left(A_1 + A_2\right)\right)$$

Once angle 3 has been found, the remainder of the triangle is solved as Angle, Side, Angle (see page 8 for the equations) to determine the other missing sides.

side 1, side 2, angle 2

TWO SIDES AND THE FOLLOWING ANGLE KNOWN has two possible solutions. When this configuration is used, both solutions may be output. The second solution will not necessarily show the parts in the same order as the input.

The other two angles are calculated with the equations shown below, and the remaining side is calculated as an Angle, Side Angle configuration.

$$A_3 = \sin^{-1}\left[\frac{S_2}{S_1}\sin A_2\right] \qquad A_1 = \cos^{-1}\left[-\cos\left(A_2 + A_3\right)\right]$$



keystrokes:



A new prompt, 2nd Solution will appear in the display after output of the first solution. If you want output of the second solution, stroke R/S If not, stroke EXIT to leave the program, or MORE.

output:

```
Side 1 = 5.7450
Angle 1 = 84°28'22.7"

Side 2 = 8.3640
Angle 2 = 36°12'30.0"

Side 3 = 9.6802
Angle 3 = 59°19'07.3"

AREA = 23.9139
```

display prompt: 2nd Solution
keystroke: R/S

```
 2nd Solution
Side 1 = 8.3640
Angle 1 = 36°12'30.0"

Side 2 = 3.8172
Angle 2 = 120°40'52.7"

Side 3 = 5.7450
Angle 3 = 23°06'37.3"

AREA = 9.4301
```

angle 3, side 1, angle 1

TWO ANGLES AND THE INCLUDED SIDE ARE KNOWN.

$$S_2 = S_1 \frac{\sin A_3}{\sin A_2}$$

$$S_3 = S_1 \cos A_3 + S_2 \cos A_2$$

$$A_2 = \cos^{-1}(-\cos(A_3 + A_1))$$

This configuration is solved by using the equations shown to the left.

The Angle, Side, Angle routine has also been used as a secondary solution to some of the other routines, after the problem has first been reduced to these three known parts.

keystrokes:

5 9 . 1 9 1 2 ANGL

5 . 7 4 5 SIDE

8 4 . 2 8 1 8 ANGL

output:

```
Side 1 = 5.7450
Angle 1 = 84°28'18.0"

Side 2 = 8.3641
Angle 2 = 36°12'30.0"

Side 3 = 9.6801
Angle 3 = 59°19'12.0"

AREA = 23.9142
```



area, side 1, angle 1

THE AREA, ONE SIDE AND ONE ANGLE KNOWN is the first of the three routines in this program which allow the area to be used as one of the known parts. *Whenever the area **is** one of the parts, it is input first.*

The equation

$$S_2 = \frac{2\ AREA}{S_1 \sin A_1}$$

is used first to reduce the problem for solution as Side 1, Angle 1, Side 2.



keystrokes:



output:

```
Side 1 = 9.6800
Angle 1 = 59°19'12.0"

Side 2 = 5.7449
Angle 2 = 84°28'18.0"

Side 3 = 8.3640
Angle 3 = 36°12'30.0"

AREA = 23.9135
```

area, angle 3, angle 1



AREA AND TWO ANGLES KNOWN is first solved for the included side, and then solved as Angle, Side, Angle. The first angle input is treated as Angle 3, the second as Angle 1. The equation used for finding Side 1 is

$$S_1 = \sqrt{\frac{2\sin A_2(\text{AREA})}{\sin A_1 \sin A_3}} \text{ where } A_2 = \cos^{-1}(-\cos(A_1 + A_3))$$

keystrokes:

23.9135 AREA

59.1912 ANGL

84.2818 ANGL

output:

```
Side 1 = 5.7449
Angle 1 = 84°28'18.0"

Side 2 = 8.3640
Angle 2 = 36°12'30.0"

Side 3 = 9.6800
Angle 3 = 59°19'12.0"

AREA = 23.9135
```

area, side 1, side 2

AREA AND TWO SIDES KNOWN is another problem which has two possible solutions.

We first find Angle 1 with the equation

$$A_1 = \sin^{-1}\left[\frac{2AREA}{S_1\ S_2}\right]$$

and then solve as Side, Angle, Side. The second solution is possible where Angle 1 may also be equal to 180° - Angle 1. This value is substituted and the second solution is output.



keystroke:



A new prompt, 2nd Solution appears in the display after output of the first solution., If you want output of the second solution, stroke R/S

If not, stroke EXIT to leave the program, or MORE to solve another triangle.

output:

```
Side 1 = 9.6800
Angle 1 = 59°19'07.1"

Side 2 = 5.7450
Angle 2 = 84°28'20.6"

Side 3 = 8.3639
Angle 3 = 36°12'32.3"

AREA = 23.9135
```

display prompt: 2nd Solution
keystroke: R/S

```
 2nd Solution
Side 1 = 9.6800
Angle 1 = 120°40'52.9"

Side 2 = 5.7450
Angle 2 = 37°55'29.4"

Side 3 = 13.5448
Angle 3 = 21°23'37.7"

AREA = 23.9135
```

The Most Commonly Asked Questions

The following questions and answers were compiled from the calls and letters we've received in the past 4+ years that we've been publishing solution books for the HP42S calculator, and are included here in the event that your question is one of them.

Q: *How do you type in the **END**?*
A: There are a number of ways . . . one easy way is to stroke XEQ ENTER and type it in, using the alpha keys. Because you stroked XEQ first, the calculator will recognize that this in not an alpha input, and substitute the actual function when you stroke ENTER again. You may input *any* function by this method.

You may also take advantage of the built-in *function catalog*, stroke ☐ + (catalog), and then the FCN menu key. You may scroll up or down with the ▲ or ▼ keys, and ***all*** of the calculator's functions are in there. When you reach the one you want, just stroke the key under the menu item.

Q: *How do you type in the **indirect** calls, such as step 12 in the first program on page 2?*
A: The indirect calls are made by stroking . . In the case of the call above, first stroke ☐ 6 (flags), then CF to bring up the prompt CF__ , then stroke . . Some of the indirect calls give a secondary prompt, requiring another . .

Q: *How do I type in a **ARCL** command?*
A: Enter ***alpha mode*** before stroking RCL or STO .

$10.00 U.S.



ISBN 0-944889-13-1